

Code Generation for a Simple First-Order Prover

Jørgen Villadsen, Anders Schlichtkrull, and Andreas Halkjær From

DTU Compute, Technical University of Denmark, 2800 Kongens Lyngby, Denmark

Abstract. We present Standard ML code generation in Isabelle/HOL of a sound and complete prover for first-order logic, taking formalizations by Tom Ridge and others as the starting point. We also define a set of so-called unfolding rules and show how to use these as a simple prover, with the aim of using the approach for teaching logic and verification to computer science students at the bachelor level.

Keywords: Isabelle/HOL, first-order logic, soundness, completeness, code generation, unfolding rules, Standard ML, Isabelle/ML

1 Introduction

We present code generation in Isabelle/HOL of a simple prover for first-order logic. We consider only Standard ML as the target programming language but OCaml, Haskell and Scala are possible too. Our work is based on Tom Ridge's entry in the Archive of Formal Proofs [6] which is building on James Margetson's formalization of work by Wainer and Wallen. Tom Ridge considers a manual translation to OCaml only. The entry also contains a recent manual translation to Standard ML by Jørgen Villadsen.

In addition to code generation we have reworked and simplified the formalization using Isar proofs instead of apply style. In particular the primitive recursive functions have been totally reworked. Our motivation is to use the approach for teaching logic and verification to computer science students at the bachelor level. However, the formalization needs further polishing, in particular where the original proofs are rather complicated — and in a few places the original proofs even have the “FIXME” comment. The formalization with a couple of examples and extra features is about 1800 lines including blank lines but excluding comments. All in all it takes less than 15 seconds to load on a fairly standard computer and is available online [9].

We first briefly describe our formalization of the proof system up to the soundness and completeness of the prover. We then explain the code generation in Standard ML and also the elegant code reflection feature using Isabelle/ML. Finally we define a set of so-called unfolding rules and show how to use these as a simple prover.

2 Proof System

We consider the same proof system as Tom Ridge [6] with a sequent as a list of formulas in negation normal form (nnf) and with no constants or functions:

$$\frac{}{\vdash P(v_{i_1}, \dots, v_{i_k}), \Gamma, \overline{P}(v_{i_1}, \dots, v_{i_k}), \Delta} Ax \quad \text{Leaf of the derivation tree.}$$

$$\frac{}{\vdash \overline{P}(v_{i_1}, \dots, v_{i_k}), \Gamma, P(v_{i_1}, \dots, v_{i_k}), \Delta} \overline{Ax} \quad \text{Leaf of the derivation tree.}$$

$$\frac{\vdash \Gamma, P(v_{i_1}, \dots, v_{i_k})}{\vdash P(v_{i_1}, \dots, v_{i_k}), \Gamma} NoAx$$

$$\frac{\vdash \Gamma, \overline{P}(v_{i_1}, \dots, v_{i_k})}{\vdash \overline{P}(v_{i_1}, \dots, v_{i_k}), \Gamma} No\overline{Ax}$$

$$\frac{\vdash \Gamma, A, B}{\vdash A \vee B, \Gamma} \vee$$

$$\frac{\vdash \Gamma, A \quad \vdash \Gamma, B}{\vdash A \wedge B, \Gamma} \wedge$$

The only branching rule.

$$\frac{\vdash \Gamma, [v_i/x]A, (\exists x.A)^{i+1}}{\vdash (\exists x.A)^i, \Gamma} \exists$$

Superscripts are only relevant for this rule, and allow $[v_i/x]A$ to be instantiated for all i .

$$\frac{\vdash \Gamma, [v_r/x]A}{\vdash \forall x.A, \Gamma} \forall$$

v_r is a fresh free variable, chosen as $r = \max(S) + 1$, where S is the set of subscripts already used for the free variables in A ($r = 0$ if there are no free variables in A).

3 Example

Formulas are in negation normal form (signed predicates and de Bruijn indices):

```
datatype nnf =
  Pre bool id (nat list) |
  Con nnf nnf |
  Dis nnf nnf |
  Uni nnf |
  Exi nnf
```

As a small test example we consider the following classically valid formula:

proposition $(\forall x. \neg P x \wedge \neg Q x) \vee (\exists x. Q x) \vee (\exists x. P x)$

We take the predicate *id* as a type synonym for *nat* and define the test example:

abbreviation (*input*) *P-id* $\equiv 0$

abbreviation (*input*) *Q-id* $\equiv \text{Suc } 0$

definition

```
test  $\equiv \text{Dis}$ 
  (Uni (Con (Pre False P-id [0]) (Pre False Q-id [0])))
  (Dis (Exi (Pre True Q-id [0])) (Exi (Pre True P-id [0])))
```

Using code generation we run the prover on the test example (*True* returned):

value *check test*

4 Formalization of Proof System

We define a function *fv* returning the free variables in a formula:

```
primrec extend :: nat list  $\Rightarrow$  nat  $\Rightarrow$  nat list where
  extend l 0 = l |
  extend l (Suc n) = n # l
```

```
primrec adjust :: nat list  $\Rightarrow$  nat list where
  adjust [] = [] |
  adjust (h # t) = extend (adjust t) h
```

```
primrec fv :: nnf  $\Rightarrow$  nat list where
  fv (Pre - - v) = v |
  fv (Con p q) = fv p @ fv q |
  fv (Dis p q) = fv p @ fv q |
  fv (Uni p) = adjust (fv p) |
  fv (Exi p) = adjust (fv p)
```

We define a function *sv* for substitution in a formula:

primrec *increase* :: (nat \Rightarrow nat) \Rightarrow nat \Rightarrow nat **where**
increase - 0 = 0 |
increase f (Suc n) = Suc (f n)

primrec *sv* :: (nat \Rightarrow nat) \Rightarrow nnf \Rightarrow nnf **where**
sv f (Pre b i v) = Pre b i (map f v) |
sv f (Con p q) = Con (sv f p) (sv f q) |
sv f (Dis p q) = Dis (sv f p) (sv f q) |
sv f (Uni p) = Uni (sv (increase f) p) |
sv f (Exi p) = Exi (sv (increase f) p)

We define a function *inst* for instantiation in a formula:

primrec *bind* :: nat \Rightarrow nat \Rightarrow nat **where**
bind x 0 = x |
bind - (Suc n) = n

definition *inst* :: nnf \Rightarrow nat \Rightarrow nnf **where**
inst p x \equiv sv (bind x) p

We define a function *fresh* returning a new variable given a list of used variables:

primrec *dec* :: nat \Rightarrow nat **where**
dec 0 = 0 |
dec (Suc n) = n

primrec *sub* :: nat \Rightarrow nat \Rightarrow nat **where**
sub x 0 = x |
sub x (Suc n) = dec (sub x n)

primrec *add* :: nat \Rightarrow nat \Rightarrow nat **where**
add x 0 = x |
add x (Suc n) = Suc (add x n)

primrec *fresh* :: nat list \Rightarrow nat **where**
fresh [] = 0 |
fresh (h # t) = Suc (add (sub (dec (fresh t)) h) h)

We define the auxiliary functions *stop* and *maps*:

primrec *stop* :: 'a list \Rightarrow 'b \Rightarrow 'b list \Rightarrow 'a list **where**
stop c - [] = c |
stop c p (h # t) = (if p = h then [] else stop c p t)

definition *maps* :: ('a \Rightarrow 'b list) \Rightarrow 'a list \Rightarrow 'b list **where**
maps f l \equiv concat (map f l)

We define the main functions *track* and *solve* on sequents:

type-synonym *sequent* = (nat \times nnf) list

```

primrec track :: sequent  $\Rightarrow$  nat  $\Rightarrow$  nnf  $\Rightarrow$  sequent list where
  track s - (Pre b i v) = stop [s @ [(0,Pre b i v)]] (Pre ( $\neg$  b) i v) (map snd s) |
  track s - (Con p q) = [s @ [(0,p)],s @ [(0,q)]] |
  track s - (Dis p q) = [s @ [(0,p),(0,q)]] |
  track s - (Uni p) = [s @ [(0,inst p (fresh (maps fv (Uni p # map snd s))))]] |
  track s n (Exi p) = [s @ [(0,inst p n),(Suc n,Exi p)]]

```

```

primrec solve :: sequent  $\Rightarrow$  sequent list where
  solve [] = [[]] |
  solve (h # t) = track t (fst h) (snd h)

```

We separate the prover algorithm in three parts (see function *main* below):

```

type-synonym 'a algorithm = ('a  $\Rightarrow$  bool)  $\Rightarrow$  ('a  $\Rightarrow$  'a)  $\Rightarrow$  'a  $\Rightarrow$  bool

```

The function *null* terminates the prover algorithm:

```

primrec null :: 'a list  $\Rightarrow$  bool where
  null [] = True |
  null (- # -) = False

```

```

definition main :: sequent list algorithm  $\Rightarrow$  nnf  $\Rightarrow$  bool where
  main a p  $\equiv$  a null (maps solve) [[(0,p)]]

```

Note the existential quantifier in the *iterator* function:

```

primrec repeat :: ('a  $\Rightarrow$  'a)  $\Rightarrow$  'a  $\Rightarrow$  nat  $\Rightarrow$  'a where
  repeat - c 0 = c |
  repeat f c (Suc n) = repeat f (f c) n

```

```

definition iterator :: 'a algorithm where
  iterator g f c  $\equiv$   $\exists$  n. g (repeat f c n)

```

Finally we obtain the prover itself:

```

definition check :: nnf  $\Rightarrow$  bool where
  check  $\equiv$  main iterator

```

5 Soundness & Completeness

For the semantics we consider only a countable universe using unit lists:

```

type-synonym proxy = unit list

```

```

type-synonym model = proxy set  $\times$  (id  $\Rightarrow$  proxy list  $\Rightarrow$  bool)

```

```

type-synonym environment = nat  $\Rightarrow$  proxy

```

```

definition is-model-environment :: model  $\Rightarrow$  environment  $\Rightarrow$  bool where
  is-model-environment m e  $\equiv$   $\forall$  n. e n  $\in$  fst m

```

```

primrec semantics :: model  $\Rightarrow$  environment  $\Rightarrow$  nnf  $\Rightarrow$  bool where
  semantics m e (Pre b i v) = (b = snd m i (map e v)) |
  semantics m e (Con p q) = (semantics m e p  $\wedge$  semantics m e q) |
  semantics m e (Dis p q) = (semantics m e p  $\vee$  semantics m e q) |
  semantics m e (Uni p) = ( $\forall z \in$  fst m.
    semantics m ( $\lambda x$ . case x of 0  $\Rightarrow$  z | Suc n  $\Rightarrow$  e n) p) |
  semantics m e (Exi p) = ( $\exists z \in$  fst m.
    semantics m ( $\lambda x$ . case x of 0  $\Rightarrow$  z | Suc n  $\Rightarrow$  e n) p)

```

The omitted soundness and completeness proof is about one thousand lines:

```

theorem check p = ( $\forall m e$ . is-model-environment m e  $\longrightarrow$  semantics m e p)

```

6 Code Generation

We must first prove a suitable “code” lemma for the *iterator* function:

```

lemma iterator[code]: iterator g f c = (if g c then True else iterator g f (f c))

```

Then the code generation is possible:

```

proposition check test

```

The simplifier can use the code equations of the underlying program:

```

  by code_simp

```

Testing using the code generator:

```

  by eval

```

Testing using normalization by evaluation (nbe):

```

  by normalization

```

Partially symbolic evaluation is possible with normalization by evaluation:

```

proposition check (Dis (Pre b i v) (Pre ( $\neg$  b) i v))

```

Finally we export the program to Standard ML (the name of a file can be added):

```

export-code check test in SML module-name SimPro

```

7 Code Reflection

The program can alternatively be compiled into the system run-time (Isabelle/ML):

```

code-reflect X datatypes nnf = Pre | Con | Dis | Uni | Exi
and nat = 0::nat | Suc
functions check

```

Here is the test example again:

```

ML {*

  val true = X.check (
    X.Dis (X.Uni (X.Con (
      X.Pre (false,X.Zero_nat,[X.Zero_nat]),
      X.Pre (false,X.Suc X.Zero_nat,[X.Zero_nat])),
    X.Dis (
      X.Exi (X.Pre (true,X.Suc X.Zero_nat,[X.Zero_nat])),
      X.Exi (X.Pre (true,X.Zero_nat,[X.Zero_nat])))))
  *)

```

8 Unfolding Rules

In the appendix we list the so-called unfolding rules using the shorthand:

abbreviation $(input) \text{ PROVER} \equiv \text{iterator null (maps solve)}$

The following simplification rules constitute the program (*check*):

theorem program:

$$\begin{aligned}
\bigwedge p. \text{check } p &\equiv \text{PROVER } [(0,p)] \\
\bigwedge h t. \text{PROVER } (h \# t) &\equiv \text{PROVER } (\text{maps solve } (h \# t)) \\
\text{PROVER } [] &\equiv \text{True} \\
\text{solve } [] &\equiv [[]] \\
\bigwedge h t. \text{solve } (h \# t) &\equiv \text{track } t \text{ (fst } h) \text{ (snd } h) \\
\bigwedge s n b i v. \text{track } s n \text{ (Pre } b i v) &\equiv \text{stop } [s @ [(0, \text{Pre } b i v)]] \text{ (Pre } (\neg b) i v) \\
(\text{map snd } s) & \\
\bigwedge s n p q. \text{track } s n \text{ (Con } p q) &\equiv [s @ [(0,p)], s @ [(0,q)]] \\
\bigwedge s n p q. \text{track } s n \text{ (Dis } p q) &\equiv [s @ [(0,p), (0,q)]] \\
\bigwedge s n p. \text{track } s n \text{ (Uni } p) &\equiv [s @ [(0, \text{inst } p \text{ (fresh (maps fv (Uni } p \# \text{map} \\
\text{snd } s)))] \\
\bigwedge s n p. \text{track } s n \text{ (Exi } p) &\equiv [s @ [(0, \text{inst } p n), (\text{Suc } n, \text{Exi } p)]] \\
\bigwedge f l. \text{maps } f l &\equiv \text{concat (map } f l) \\
\bigwedge c p. \text{stop } c p [] &\equiv c \\
\bigwedge c p h t. \text{stop } c p \text{ (} h \# t) &\equiv (\text{if } p = h \text{ then } [] \text{ else stop } c p t) \\
\text{fresh } [] &\equiv 0 \\
\bigwedge h t. \text{fresh } (h \# t) &\equiv \text{Suc (add (sub (dec (fresh } t)) h) h) \\
\bigwedge x. \text{add } x 0 &\equiv x \\
\bigwedge x n. \text{add } x \text{ (Suc } n) &\equiv \text{Suc (add } x n) \\
\bigwedge x. \text{sub } x 0 &\equiv x \\
\bigwedge x n. \text{sub } x \text{ (Suc } n) &\equiv \text{dec (sub } x n) \\
\text{dec } 0 &\equiv 0 \\
\bigwedge n. \text{dec (Suc } n) &\equiv n
\end{aligned}$$

$$\begin{aligned}
&\wedge p x. \text{inst } p x \equiv \text{sv } (\text{bind } x) p \\
&\wedge x. \text{bind } x 0 \equiv x \\
&\wedge x n. \text{bind } x (\text{Suc } n) \equiv n \\
&\wedge f b i v. \text{sv } f (\text{Pre } b i v) \equiv \text{Pre } b i (\text{map } f v) \\
&\wedge f p q. \text{sv } f (\text{Con } p q) \equiv \text{Con } (\text{sv } f p) (\text{sv } f q) \\
&\wedge f p q. \text{sv } f (\text{Dis } p q) \equiv \text{Dis } (\text{sv } f p) (\text{sv } f q) \\
&\wedge f p. \text{sv } f (\text{Uni } p) \equiv \text{Uni } (\text{sv } (\text{increase } f) p) \\
&\wedge f p. \text{sv } f (\text{Exi } p) \equiv \text{Exi } (\text{sv } (\text{increase } f) p) \\
&\wedge f. \text{increase } f 0 \equiv 0 \\
&\wedge f n. \text{increase } f (\text{Suc } n) \equiv \text{Suc } (f n) \\
&\wedge b i v. \text{fv } (\text{Pre } b i v) \equiv v \\
&\wedge p q. \text{fv } (\text{Con } p q) \equiv \text{fv } p @ \text{fv } q \\
&\wedge p q. \text{fv } (\text{Dis } p q) \equiv \text{fv } p @ \text{fv } q \\
&\wedge p. \text{fv } (\text{Uni } p) \equiv \text{adjust } (\text{fv } p) \\
&\wedge p. \text{fv } (\text{Exi } p) \equiv \text{adjust } (\text{fv } p) \\
&\text{adjust } [] \equiv [] \\
&\wedge h t. \text{adjust } (h \# t) \equiv \text{extend } (\text{adjust } t) h \\
&\wedge l. \text{extend } l 0 \equiv l \\
&\wedge l n. \text{extend } l (\text{Suc } n) \equiv n \# l
\end{aligned}$$

The following simplification rules depends on the datatype (*nmf*):

theorem data:

$$\begin{aligned}
&\wedge b i v p q. \text{Pre } b i v = \text{Con } p q \equiv \text{False} \\
&\wedge b i v p q. \text{Con } p q = \text{Pre } b i v \equiv \text{False} \\
&\wedge b i v p q. \text{Pre } b i v = \text{Dis } p q \equiv \text{False} \\
&\wedge b i v p q. \text{Dis } p q = \text{Pre } b i v \equiv \text{False} \\
&\wedge b i v p. \text{Pre } b i v = \text{Uni } p \equiv \text{False} \\
&\wedge b i v p. \text{Uni } p = \text{Pre } b i v \equiv \text{False} \\
&\wedge b i v p. \text{Pre } b i v = \text{Exi } p \equiv \text{False} \\
&\wedge b i v p. \text{Exi } p = \text{Pre } b i v \equiv \text{False} \\
&\wedge p q p' q'. \text{Con } p q = \text{Dis } p' q' \equiv \text{False} \\
&\wedge p q p' q'. \text{Dis } p' q' = \text{Con } p q \equiv \text{False} \\
&\wedge p q p'. \text{Con } p q = \text{Uni } p' \equiv \text{False} \\
&\wedge p q p'. \text{Uni } p' = \text{Con } p q \equiv \text{False} \\
&\wedge p q p'. \text{Con } p q = \text{Exi } p' \equiv \text{False} \\
&\wedge p q p'. \text{Exi } p' = \text{Con } p q \equiv \text{False} \\
&\wedge p q p'. \text{Dis } p q = \text{Uni } p' \equiv \text{False} \\
&\wedge p q p'. \text{Uni } p' = \text{Dis } p q \equiv \text{False} \\
&\wedge p q p'. \text{Dis } p q = \text{Exi } p' \equiv \text{False} \\
&\wedge p q p'. \text{Exi } p' = \text{Dis } p q \equiv \text{False} \\
&\wedge p p'. \text{Uni } p = \text{Exi } p' \equiv \text{False} \\
&\wedge p p'. \text{Exi } p' = \text{Uni } p \equiv \text{False} \\
&\wedge b i v b' i' v'. \text{Pre } b i v = \text{Pre } b' i' v' \equiv b = b' \wedge i = i' \wedge v = v' \\
&\wedge p q p' q'. \text{Con } p q = \text{Con } p' q' \equiv p = p' \wedge q = q' \\
&\wedge p q p' q'. \text{Dis } p q = \text{Dis } p' q' \equiv p = p' \wedge q = q' \\
&\wedge p p'. \text{Uni } p = \text{Uni } p' \equiv p = p' \\
&\wedge p p'. \text{Exi } p = \text{Exi } p' \equiv p = p'
\end{aligned}$$

The following simplification rules provide a functional programming language:

theorem *library*:

$$\begin{aligned}
\bigwedge f. \text{map } f \ [] &\equiv [] \\
\bigwedge f h t. \text{map } f (h \# t) &\equiv f h \# \text{map } f t \\
\text{concat } [] &\equiv [] \\
\bigwedge h t. \text{concat } (h \# t) &\equiv h @ \text{concat } t \\
\bigwedge l. [] @ l &\equiv l \\
\bigwedge h t l. (h \# t) @ l &\equiv h \# t @ l \\
\bigwedge x y. \text{if True then } x \text{ else } y &\equiv x \\
\bigwedge x y. \text{if False then } x \text{ else } y &\equiv y \\
\neg \text{True} &\equiv \text{False} \\
\neg \text{False} &\equiv \text{True} \\
\bigwedge b. \neg \neg b &\equiv b \\
\bigwedge x y. \text{fst } (x,y) &\equiv x \\
\bigwedge x y. \text{snd } (x,y) &\equiv y \\
\bigwedge n. 0 = \text{Suc } n &\equiv \text{False} \\
\bigwedge n. \text{Suc } n = 0 &\equiv \text{False} \\
\bigwedge h t. [] = h \# t &\equiv \text{False} \\
\bigwedge h t. h \# t = [] &\equiv \text{False} \\
\text{True} = \text{False} &\equiv \text{False} \\
\text{False} = \text{True} &\equiv \text{False} \\
0 = 0 &\equiv \text{True} \\
[] = [] &\equiv \text{True} \\
\text{True} = \text{True} &\equiv \text{True} \\
\text{False} = \text{False} &\equiv \text{True} \\
\bigwedge x y x' y'. (x,y) = (x',y') &\equiv x = x' \wedge y = y' \\
\bigwedge n n'. \text{Suc } n = \text{Suc } n' &\equiv n = n' \\
\bigwedge h t h' t'. h \# t = h' \# t' &\equiv h = h' \wedge t = t' \\
\bigwedge b. \text{True} \wedge b &\equiv b \\
\bigwedge b. \text{False} \wedge b &\equiv \text{False}
\end{aligned}$$

Let us consider the test example (without the *test* abbreviation):

lemma *check*

$$\begin{aligned}
&(\text{Dis } (\text{Uni } (\text{Con } (\text{Pre False } 0 \ [0]) \ (\text{Pre False } (\text{Suc } 0) \ [0]))) \\
&(\text{Dis } (\text{Exi } (\text{Pre True } (\text{Suc } 0) \ [0]) \ (\text{Exi } (\text{Pre True } 0 \ [0])))
\end{aligned}$$

We use just the unfolding rules:

$$\begin{aligned}
&\text{unfolding program data library} \\
&\text{by (rule TrueI)}
\end{aligned}$$

We can even “single-step” it (here with the *test* abbreviation):

proposition *check test*

```

unfolding test_def
unfolding program(1)
unfolding program(2)
unfolding program(3-) data library
unfolding program(2)
unfolding program(3-) data library
unfolding program(2)
unfolding program(3-) data library
unfolding program(2)
unfolding program(3-) data library
unfolding program(2)
unfolding program(3-) data library
unfolding program(2)
unfolding program(3-) data library
unfolding program(2)
unfolding program(3-) data library
by (rule TrueI)

```

Here are the results for the main steps (using the *PROVER* abbreviation):

check test

check

```

(Dis (Uni (Con (Pre False 0 [0]) (Pre False (Suc 0) [0])))
  (Dis (Exi (Pre True (Suc 0) [0])) (Exi (Pre True 0 [0]))))

```

PROVER

```

[[ (0, Dis (Uni (Con (Pre False 0 [0]) (Pre False (Suc 0) [0])))
  (Dis (Exi (Pre True (Suc 0) [0])) (Exi (Pre True 0 [0]))))
]]

```

PROVER

```

[[ (0, Uni (Con (Pre False 0 [0]) (Pre False (Suc 0) [0])),
  (0, Dis (Exi (Pre True (Suc 0) [0])) (Exi (Pre True 0 [0]))))
]]

```

PROVER

```

[[ (0, Dis (Exi (Pre True (Suc 0) [0])) (Exi (Pre True 0 [0])),
  (0, Con (Pre False 0 [0]) (Pre False (Suc 0) [0]))
]]

```

PROVER

```

[[ (0, Con (Pre False 0 [0]) (Pre False (Suc 0) [0])),
  (0, Exi (Pre True (Suc 0) [0])), (0, Exi (Pre True 0 [0]))
]]

```

PROVER

$[[(0, \text{Exi } (\text{Pre True } (\text{Suc } 0) [0])), (0, \text{Exi } (\text{Pre True } 0 [0])),$
 $(0, \text{Pre False } 0 [0]),$
 $(0, \text{Exi } (\text{Pre True } (\text{Suc } 0) [0])), (0, \text{Exi } (\text{Pre True } 0 [0])),$
 $(0, \text{Pre False } (\text{Suc } 0) [0])]]$

PROVER

$[[(0, \text{Exi } (\text{Pre True } 0 [0])), (0, \text{Pre False } 0 [0]),$
 $(0, \text{Pre True } (\text{Suc } 0) [0]),$
 $(\text{Suc } 0, \text{Exi } (\text{Pre True } (\text{Suc } 0) [0])),$
 $(0, \text{Exi } (\text{Pre True } 0 [0])), (0, \text{Pre False } (\text{Suc } 0) [0]),$
 $(0, \text{Pre True } (\text{Suc } 0) [0]),$
 $(\text{Suc } 0, \text{Exi } (\text{Pre True } (\text{Suc } 0) [0]))]]$

PROVER

$[[(0, \text{Pre False } 0 [0]), (0, \text{Pre True } (\text{Suc } 0) [0]),$
 $(\text{Suc } 0, \text{Exi } (\text{Pre True } (\text{Suc } 0) [0])), (0, \text{Pre True } 0 [0]),$
 $(\text{Suc } 0, \text{Exi } (\text{Pre True } 0 [0])),$
 $(0, \text{Pre False } (\text{Suc } 0) [0]), (0, \text{Pre True } (\text{Suc } 0) [0]),$
 $(\text{Suc } 0, \text{Exi } (\text{Pre True } (\text{Suc } 0) [0])), (0, \text{Pre True } 0 [0]),$
 $(\text{Suc } 0, \text{Exi } (\text{Pre True } 0 [0]))]]$

True

9 Related Work and Conclusion

There are several formalizations of proof systems in Isabelle/HOL and other systems [1–5, 7, 8]. The present simple prover based on Tom Ridge’s entry in the Archive of Formal Proofs [6] seems indeed unique given its overall simplicity, full automation and code generation. Future work include further polishing the proofs and development of a stand-alone tool for illustrating the unfolding rules.

Acknowledgement. Many thanks to Tom Ridge, Alexander Birch Jensen and the Isabelle team at TUM for help and discussions.

Appendix: Simple Prover Unfolding Rules*The “program” rules*

$$\text{check } p \equiv \text{PROVER } [[(0,p)]]$$

$$\text{PROVER } (h \# t) \equiv \text{PROVER } (\text{maps solve } (h \# t))$$

$$\text{PROVER } [] \equiv \text{True}$$

$$\text{solve } [] \equiv [[]]$$

$$\text{solve } (h \# t) \equiv \text{track } t \text{ (fst } h \text{) (snd } h \text{)}$$

$$\text{track } s \ n \ (\text{Pre } b \ i \ v) \equiv \text{stop } [s \ @ \ [(0,\text{Pre } b \ i \ v)]] \ (\text{Pre } (\neg b) \ i \ v) \ (\text{map } \text{snd } s)$$

$$\text{track } s \ n \ (\text{Con } p \ q) \equiv [s \ @ \ [(0,p)], s \ @ \ [(0,q)]]$$

$$\text{track } s \ n \ (\text{Dis } p \ q) \equiv [s \ @ \ [(0,p), (0,q)]]$$

$$\text{track } s \ n \ (\text{Uni } p) \equiv [s \ @ \ [(0,\text{inst } p \ (\text{fresh } (\text{maps } \text{fv } (\text{Uni } p \ \# \ \text{map } \text{snd } s))))]]$$

$$\text{track } s \ n \ (\text{Exi } p) \equiv [s \ @ \ [(0,\text{inst } p \ n), (\text{Suc } n, \text{Exi } p)]]$$

$$\text{maps } f \ l \equiv \text{concat } (\text{map } f \ l)$$

$$\text{stop } c \ p \ [] \equiv c$$

$$\text{stop } c \ p \ (h \# t) \equiv (\text{if } p = h \text{ then } [] \text{ else } \text{stop } c \ p \ t)$$

$$\text{fresh } [] \equiv 0$$

$$\text{fresh } (h \# t) \equiv \text{Suc } (\text{add } (\text{sub } (\text{dec } (\text{fresh } t)) \ h) \ h)$$

$$\text{add } x \ 0 \equiv x$$

$$\text{add } x \ (\text{Suc } n) \equiv \text{Suc } (\text{add } x \ n)$$

$$\text{sub } x \ 0 \equiv x$$

$$\text{sub } x \ (\text{Suc } n) \equiv \text{dec } (\text{sub } x \ n)$$

$$\text{dec } 0 \equiv 0$$

$$\text{dec } (\text{Suc } n) \equiv n$$

$$\text{inst } p \ x \equiv \text{sv } (\text{bind } x) \ p$$

$$\text{bind } x \ 0 \equiv x$$

$$\text{bind } x \ (\text{Suc } n) \equiv n$$

$$\text{sv } f \ (\text{Pre } b \ i \ v) \equiv \text{Pre } b \ i \ (\text{map } f \ v)$$

$$\text{sv } f \ (\text{Con } p \ q) \equiv \text{Con } (\text{sv } f \ p) \ (\text{sv } f \ q)$$

$$\text{sv } f \ (\text{Dis } p \ q) \equiv \text{Dis } (\text{sv } f \ p) \ (\text{sv } f \ q)$$

$$\text{sv } f \ (\text{Uni } p) \equiv \text{Uni } (\text{sv } (\text{increase } f) \ p)$$

$$\text{sv } f \ (\text{Exi } p) \equiv \text{Exi } (\text{sv } (\text{increase } f) \ p)$$

$increase\ f\ 0 \equiv 0$
 $increase\ f\ (Suc\ n) \equiv Suc\ (f\ n)$

$fv\ (Pre\ b\ i\ v) \equiv v$
 $fv\ (Con\ p\ q) \equiv fv\ p\ @\ fv\ q$
 $fv\ (Dis\ p\ q) \equiv fv\ p\ @\ fv\ q$
 $fv\ (Uni\ p) \equiv adjust\ (fv\ p)$
 $fv\ (Exi\ p) \equiv adjust\ (fv\ p)$

$adjust\ [] \equiv []$
 $adjust\ (h\ \#t) \equiv extend\ (adjust\ t)\ h$

$extend\ l\ 0 \equiv l$
 $extend\ l\ (Suc\ n) \equiv n\ \#l$

The “data” rules

$Pre\ b\ i\ v = Con\ p\ q \equiv False$
 $Con\ p\ q = Pre\ b\ i\ v \equiv False$
 $Pre\ b\ i\ v = Dis\ p\ q \equiv False$
 $Dis\ p\ q = Pre\ b\ i\ v \equiv False$
 $Pre\ b\ i\ v = Uni\ p \equiv False$
 $Uni\ p = Pre\ b\ i\ v \equiv False$
 $Pre\ b\ i\ v = Exi\ p \equiv False$
 $Exi\ p = Pre\ b\ i\ v \equiv False$
 $Con\ p\ q = Dis\ p'\ q' \equiv False$
 $Dis\ p'\ q' = Con\ p\ q \equiv False$
 $Con\ p\ q = Uni\ p' \equiv False$
 $Uni\ p' = Con\ p\ q \equiv False$
 $Con\ p\ q = Exi\ p' \equiv False$
 $Exi\ p' = Con\ p\ q \equiv False$
 $Dis\ p\ q = Uni\ p' \equiv False$
 $Uni\ p' = Dis\ p\ q \equiv False$
 $Dis\ p\ q = Exi\ p' \equiv False$
 $Exi\ p' = Dis\ p\ q \equiv False$
 $Uni\ p = Exi\ p' \equiv False$
 $Exi\ p' = Uni\ p \equiv False$

$Pre\ b\ i\ v = Pre\ b'\ i'\ v' \equiv b = b' \wedge i = i' \wedge v = v'$
 $Con\ p\ q = Con\ p'\ q' \equiv p = p' \wedge q = q'$
 $Dis\ p\ q = Dis\ p'\ q' \equiv p = p' \wedge q = q'$
 $Uni\ p = Uni\ p' \equiv p = p'$
 $Exi\ p = Exi\ p' \equiv p = p'$

The “library” rules

$$\begin{aligned} \text{map } f [] &\equiv [] \\ \text{map } f (h \# t) &\equiv f h \# \text{map } f t \end{aligned}$$

$$\begin{aligned} \text{concat } [] &\equiv [] \\ \text{concat } (h \# t) &\equiv h @ \text{concat } t \end{aligned}$$

$$\begin{aligned} [] @ l &\equiv l \\ (h \# t) @ l &\equiv h \# t @ l \end{aligned}$$

$$\begin{aligned} \text{if } \text{True} \text{ then } x \text{ else } y &\equiv x \\ \text{if } \text{False} \text{ then } x \text{ else } y &\equiv y \end{aligned}$$

$$\begin{aligned} \neg \text{True} &\equiv \text{False} \\ \neg \text{False} &\equiv \text{True} \\ \neg \neg b &\equiv b \end{aligned}$$

$$\begin{aligned} \text{fst } (x,y) &\equiv x \\ \text{snd } (x,y) &\equiv y \end{aligned}$$

$$\begin{aligned} 0 = \text{Suc } n &\equiv \text{False} \\ \text{Suc } n = 0 &\equiv \text{False} \end{aligned}$$

$$\begin{aligned} [] = h \# t &\equiv \text{False} \\ h \# t = [] &\equiv \text{False} \end{aligned}$$

$$\begin{aligned} \text{True} = \text{False} &\equiv \text{False} \\ \text{False} = \text{True} &\equiv \text{False} \end{aligned}$$

$$0 = 0 \equiv \text{True}$$

$$[] = [] \equiv \text{True}$$

$$\begin{aligned} \text{True} = \text{True} &\equiv \text{True} \\ \text{False} = \text{False} &\equiv \text{True} \end{aligned}$$

$$(x,y) = (x',y') \equiv x = x' \wedge y = y'$$

$$\text{Suc } n = \text{Suc } n' \equiv n = n'$$

$$h \# t = h' \# t' \equiv h = h' \wedge t = t'$$

$$\begin{aligned} \text{True} \wedge b &\equiv b \\ \text{False} \wedge b &\equiv \text{False} \end{aligned}$$

References

1. Berghofer, S.: First-order logic according to Fitting. Archive of Formal Proofs (Aug 2007), <http://isa-afp.org/entries/FOL-Fitting.shtml>, Formal proof development
2. Blanchette, J.C., Popescu, A., Traytel, D.: Unified classical logic completeness: A coinductive pearl. In: Demri, S., Kapur, D., Weidenbach, C. (eds.) IJCAR 2014. LNCS, vol. 8562, pp. 46–60. Springer (2014)
3. Breitner, J., Lohner, D.: The meta theory of the incredible proof machine. Archive of Formal Proofs (May 2016), http://isa-afp.org/entries/Incredible_Proof_Machine.shtml, Formal proof development
4. Harrison, J.: Towards self-verification of HOL Light. In: Furbach, U., Shankar, N. (eds.) IJCAR 2006. LNCS, vol. 4130, pp. 177–191. Springer (2006)
5. Kumar, R., Arthan, R., Myreen, M.O., Owens, S.: Self-formalisation of higher-order logic: Semantics, soundness, and a verified implementation. *Journal of Automated Reasoning* 56(3), 221–259 (2016)
6. Ridge, T.: A mechanically verified, efficient, sound and complete theorem prover for first order logic. Archive of Formal Proofs (Sep 2004), <http://isa-afp.org/entries/Verified-Prover.shtml>, Formal proof development
7. Schlichtkrull, A.: Formalization of the resolution calculus for first-order logic. In: Blanchette, J.C., Merz, S. (eds.) ITP 2016. LNCS, vol. 9807. Springer (2016)
8. Villadsen, J., Jensen, A.B., Schlichtkrull, A.: NaDeA: A natural deduction assistant with a formalization in Isabelle. <https://nadea.compute.dtu.dk>
9. Villadsen, J., Schlichtkrull, A., From, A.H.: SimPro - Simple Prover - With a Formalization in Isabelle. <https://github.com/logic-tools/simpro>